
Objectives:

· To describe the interrupt mechanism of Pentium 

· To describe hardware, software, and processor interrupts 

· To describe Interrupt processing 

· To briefly describe BIOS interrupts 

· To briefly describe DOS interrupts 

· To illustrate the use of interrupts in a time display program 

· To describe how to write an interrupt service routine 

· To briefly describe file processing using INT 21h 

Interrupt is a mechanism by which a program's flow of control can be altered. When an interrupt occurs, the CPU suspends its execution of the current program, and transfers control to an Interrupt Service Routine (ISR), also called a Handler, that will provide the requested service by the interrupt. When the ISR is completed, the original program resumes execution as if it were not interrupted. 

Interrupts vs. Procedures 

Although the behavior of interrupts is analogous to procedures, there are some basic differences that make interrupts almost indispensable. These differences are highlighted below: 

· Interrupts can be initiated by both software and hardware; procedures can be initiated only by software. 

· Interrupt mechanism provides an efficient way to handle unanticipated events. For example, if the program goes into an infinite loop, ctrl-break could cause an interrupt to suspend the program execution. 

· Interrupt Service Routines are memory resident while procedures are loaded with application programs. 

· Interrupts are identified by numbers while procedures are identified by names. 

A Taxonomy of Interrupts  [image: image9.png]



	There are three main types of interrupts:  
	A taxonomy of Pentium interrupts is shown below:

	· Software interrupts 

· Hardware interrupts 
· Processor interrupts or Exceptions

Software Interrupts 

Software interrupts are initiated by executing the interrupt instruction INT in a program. Software interrupts are mainly used in accessing I/O devices such as the keyboard, printer, screen, disk drive, etc. For example, INT 21H instruction is used to read a character or a string from the keyboard or display a character or a string on the screen. Software interrupts can be classified into system-defined or user-defined. System-defined software interrupts are those whose interrupt service routines are supported by BIOS and DOS. User-defined interrupts are those whose interrupt service routines are provided by the user. The format of the interrupt instruction is: 
INT interrupt-type
where interrupt-type is an integer in the range 0 through 255. Thus, there are 256 different interrupt types. This is a sufficiently large number, as each interrupt type can be parameterized to provide several services. For example, there are more than 80 different services (called functions) provided by DOS through INT 21H. The following table shows which of the 256 interrupt types interrupts are BIOS interrupts, DOS interrupts, or user interrupts. 

	Interrupt Type 
	Allocation 

	0-1Fh 
	BIOS Interrupts 

	20h-3Fh 
	DOS Interrupts

	40h-7Fh 
	Reserved 

	80h-F0h 
	ROM Basic 

	F1h-FFh 
	User Defined 


In addition to the INT instruction, there is only one additional conditional software interrupt instruction, INTO, interrupt on overflow, which invokes INT 4 when the overflow flag is set. Note that the interrupt instructions have no effect on the flags. 

Hardware Interrupts 

Hardware interrupts are generated by hardware devices to get the attention of the CPU. For example, when a key is pressed, the keyboard generates an interrupt causing the CPU to suspend its present activity and execute the keyboard interrupt service routine to process the key. Hardware interrupts can be either maskable or non-maskable. Maskable interrupts are initiated through INTR pin while non-maskable interrupts are initiated through the NMI pin. The non-maskable interrupts are serviced by the CPU immediately after completing the execution of the current instruction. However, maskable interrupts can be delayed until execution reaches a convenient point. One example of non-maskable interrupts is the RAM parity error indicating memory malfunction. 

Interrupt Flag (IF) 

The Interrupt Flag (IF) controls whether maskable interrupts are delayed or not. When IF=1, the maskable interrupts will be serviced by the CPU. When IF=0 the maskable interrupts will be delayed until the IF becomes 1. The instruction STI / CLI is used to set /clear the interrupt flag. 

Identifying Hardware Interrupt Types 

In response to a hardware interrupt request on the INTR pin, the CPU initiates an interrupt acknowledge sequence. As part of this sequence, the CPU sends out an interrupt acknowledge (INTA) signal, and the interrupting device places the interrupt type number on the data bus. 

Handling Interrupts from Several I/O Devices 

Most hardware interrupts are maskable interrupts. Thus, all interrupts requested from external devices are initiated through the INTR pin. When more than one device interrupts, we have to have a mechanism to prioritize these interrupts and forward only one interrupt request at a time to the CPU while keeping the other interrupt requests pending for their turn. This mechanism can be implemented by using a special chip - the Intel 8259 Programmable Interrupt Controller. 

Processor Interrupts or Exceptions 

Exceptions are processor interrupts to handle instruction faults. An example of an exception is the divide error fault, which is generated when the quotient can not fit in the quotient register. Exceptions are classified into three types depending on the way they are reported and whether or not the instruction that interrupted is restarted:  Faults, Traps & Aborts 

Faults and traps are reported at instructions boundaries. When a fault is reported, the system state is restored to the state before the instruction that caused the interrupt so that the instruction can be restarted. Examples of faults are the divide error fault and the segment-not-present fault. The segment-not-present fault is caused by a reference to data in a segment that is not in memory. Then, the exception handler must load the missing segment from the hard disk and resume program execution starting with the instruction that caused the exception. 
On the other hand, when a trap is reported, the system state is restored to the state after the instruction that caused the interrupt. An example of a trap is the overflow exception (INT 4). Aborts are exceptions that report severe errors. Examples include hardware errors and inconsistent values in system tables. 
Interrupt Processing  [image: image11.png]



When the CPU is started, the BIOS and DOS ISRs are loaded into memory and they stay memory resident. In order to have flexibility in storing these ISRs in memory in any location, their addresses are stored in an interrupt descriptor table (IDT) or interrupt vector table (IVT). 
In protected mode the IDT can be stored at any location and its address is stored at the 48-bit register IDTR. In this case, the address of an ISR requires 8 bytes. So, the size of the IDT is 2048 bytes. However, in real mode the IDT is stored at base address 0. In this case, the address of an ISR requires 4 bytes only, 2 bytes for the IP offset and 2 bytes for the CS. Since each entry in the IVT requires 4 bytes, the interrupt type is multiplied by 4 to get the corresponding ISR pointer in the table. 
For example, the ISR pointer for INT 0 is 0000h, the pointer for INT 1 is 0004h, the pointer for INT 2 is 0008h, and the pointer for INT 21h is 0084h. For INT 21h, the IP register will be loaded with the 16-bit from addresses [0085] and [0084] and the CS register will be loaded with the 16-bit from addresses [0087] and [0086]. 
	This figure illustrates the input vector layout in memory for the real mode. When an interrupt occurs, the following action is taken by the CPU:

· Push flags register onto the stack, 

· Clear interrupt and trap flags, 

· Push CS onto the stack, 

· Push IP onto the stack, 

· Load CS with the 16-bit at memory address [interrupt-type*4+2], 

· Load IP with the 16-bit at memory address [interrupt-type*4]. 

Just like procedures, ISRs should end with a return statement to send control back to the interrupted program. The interrupt return instruction, IRET, is used for this purpose. When the IRET instruction is executed, the CPU performs the following steps: 

· Pop the 16-bit from the top of the stack into the IP register, 

· Pop the 16-bit from the top of the stack into the CS register, 

· Pop the 16-bit from the top of the stack into the flags register. 

Interrupt types 0 to 1Fh are known as BIOS interrupts: most ISRs are BIOS routines residing in the ROM. 

Interrupt Types 0-7 

Interrupt types 0-7 are reserved by Intel: types 0-5 are predefined, 6 and 7 are not yet used. 

	00h
	Divide Overflow
	Generated when a DIV or IDIV operation produces an overflow. This occurs when the quotient can not fit in the destination register. The interrupt 0 routine displays the message "DIVIDE OVERFLOW" and returns control to DOS.

	01h
	Single Step
	Single-stepping is a useful debugging tool to observe the behavior of a program instruction by instruction. This interrupt is generated when the Trap Flag (TF) is set. The ISR can be used to display relevant information about the state of the program (for example, the contents of all registers). Single-stepping is stopped by clearing TF. There are dedicated instructions for setting and clearing the TF.

	02h
	Non-Maskable Interrupt
	Interrupt 2 is the hardware interrupt that cannot be masked out by clearing the TF. The IBM PC uses this interrupt to signal memory and I/O parity errors that indicate bad chips.

	03h
	Breakpoint
	The only 1-byte interrupt instruction (opcode: CCh); other interrupt instructions are 2-byte instructions. Inserting a breakpoint in a program involves replacing the program code byte by CCh while saving the code byte for later restoration to remove the breakpoint. Breakpoints are used by program debuggers.

	04h
	Overflow
	Generated by the instruction INTO (interrupt if overflow) when the OF is set. Programmers may write their own routines to handle unexpected overflows. Executing the instruction INT 4 will invoke the ISR for this interrupt unconditionally while INTO invokes it according to OF. The overflow condition is usually detected and processed using the instructions JO and JNO not INTO.

	05h
	Print Screen
	The BIOS interrupt 5 routine sends the video screen information to the printer. An INT 5 instruction is generated by the keyboard interrupt routine (interrupt 9) when the PrtSC (print screen) key is pressed. 


Types 8h-Fh 

The 8086 has only one pin, INTR pin, for maskable hardware interrupt signals. To allow more devices to interrupt the 8086, the Intel 8259 Programmable Interrupt Controller chip is used. It can interface up to 8 devices. Interrupt types 8h-Fh are generated by hardware devices connected to the 8259. The older PCs uses only 8, 9, and Eh. 

	08h
	Timer
	The IBM PC contains a timer circuit that generates an interrupt once every 54.92 milliseconds (about 18.2 times per second). The BIOS interrupt 8 routine services the timer circuit. It uses the timer signals (ticks) to keep track of the time of the day.

	09h
	Keyboard
	Generated by the keyboard each time a key is pressed or released. It reads a scan code and stores it in the keyboard buffer. In addition, it also identifies special key combinations such ctrl-break. The keyboard buffer has the capacity to store up to 15 keys. When the buffer is full, pressing a key causes the BIOS to beep, indicating that the key stroke is lost. The keyboard controller supplies the key identity by means of a scan code. The scan code of a key is simply an identification number given to the key based on its location in the keyboard. The scan code has no relation to the ASCII code. The interrupt 9 routine receives the scan code and generates the equivalent ASCII code, if there is one. Both the scan code and the ASCII code are placed in the keyboard buffer. 

	0EH
	Diskette Error
	The BIOS interrupt Eh handles diskette errors.


Interrupt Types 10h-1Fh 

Interrupt 10h-1Fh can be called by application programs to perform various I/O operations and status checking. Associated with each I/O device there is a device controller or I/O controller that acts as a hardware interface between the processor and the I/O device. The device controller performs many of the low-level tasks specific to the I/O device. This allows the CPU to interact with the device at higher level. For each device controller, there is a software interface that provides a clean interface to access the device. This software interface is called the device driver.
	10H
	Video
	The BIOS interrupt 10h routine is the video driver.

	11H
	Equipment Check
	The BIOS interrupt 11h routine returns the equipment configuration of the particular PC. The return code is placed in register AX.

	12H
	Memory Size
	This interrupt returns (in KB) the amount of conventional memory a computer has in register AX. Conventional memory refers to memory circuits with addresses below 640KB. 

	13H
	Disk I/O
	The BIOS interrupt 13h routine is the disk driver. It allows application programs to do disk I/O.

	14H
	Communications
	The BIOS interrupt 14h routine is the communications driver that interacts with the serial ports.

	15H
	Cassette
	This interrupt was used by the original PC for cassette interface.

	16H
	Keyboard I/O
	The BIOS interrupt 16h routine is the keyboard driver. 

	17H
	Printer I/O
	The BIOS interrupt 17h routine is the printer driver.

	18H
	BASIC
	The BIOS interrupt 18h routine transfers control to ROM BASIC.

	19H
	Bootstrap
	The BIOS interrupt 19h routine reboots the system.

	1AH
	Time of Day
	The BIOS interrupt 1Ah routine allows a program to get and set the timer tick count.

	1BH
	Ctrl Break
	This interrupt is called by the INT 9 routine when the Ctrl-break key is pressed. The ISR contains only an IRET instruction. Users may write their own routine to handle the Ctrl-break key.

	1CH
	Timer Tick
	Interrupt 1Ch is called by the INT 8 routine each time the timer circuit interrupts. The ISR contains only an IRET instruction. Users may write their own service 


Interrupt 16h -- Keyboard I/O Example 
We list here some examples. To read a character from the keyboard, function 0 in AH is used as follows: 

	Function 00H -- Read a character from the keyboard
Input:

AH=00H
Returns:
if AL<>0 then


   AL = ASCII code of the key entered


   AH = Scan code of the key entered


if AL = 0


   AH = Scan code of the extended key entered


This BIOS function can be used to read a character from the keyboard. If the keyboard buffer is empty, it waits for a character to be entered. The value returned in AL determines if the key represents an ASCII character or an extended key character. In both cases, the scan code is placed in the AH register and the ASCII and scan codes are removed from the keyboard buffer. To check if keyboard buffer is empty or not, we can use function 1 as follows: 

	Function 01H -- Check keyboard buffer
Input:

AH=01H
Returns:
ZF= 1 if the keyboard buffer is empty


ZF= 0 if there is at least one character available.


In this case, the ASCII and scan codes are placed 


in the AL and AH registers as in function 00. The 


codes, however, are not removed from the buffer.


Function 2 can be used to check keyboard status with regard to shift and toggle keys. 

	Function 02H – Check keyboard status
Input:

AH=02H
Returns:
AL= status of the shift and toggle keys


The following table indicates the bit assignment for shift and toggle keys. A bit with a value of 1 indicates the presence of a condition. 
	Bits in AL
	Key Assignment 

	7
	Ins lock switch is on 

	6
	Caps lock switch is on 

	5
	Num lock switch is on 

	4
	Scroll lock switch is on 

	3
	Alt key depressed 

	2
	Ctrl key depressed 

	1
	Left shift key depressed 

	0
	Right shift key depressed 


Interrupt 17h -- Printer I/O 

The BIOS interrupt 17h routine is the printer driver. The routine supports three functions: 0-2. 

· Function 0 writes a character to a printer; input values are AH=0, AL=character to be printed, DX=printer number (0=LPT1, 1=LPT2, 3=LPT3). 

· Function 1 initializes a printer port; input values are AH=1, DX=printer number. 

· Function 2 gets printer status; input values are AH=2, DX= printer number. 

For all functions the status is returned in AH. The following table shows the meaning of the bits returned in AH: 

	Bits in AH
	Meaning 

	7
	= 1 printer not busy 

	6
	= 1 print acknowledge 

	5
	= 1 out of paper 

	4
	= 1 printer online 

	3
	= 1 I/O error 

	2
	= 1 not used 

	1
	= 1 not used 

	0
	= 1 printer timed-out 


Next, we show an example for printing character 0 on the printer. Because printers contain buffers for data, the 0 will not be printed until a carriage return or line feed character is sent. 

	Example: Print a 0 on the printer.

	
MOV AH, 0
; function 0, print character 

MOV AL, '0'
; character to be printed in AL

MOV DX, 0
; Printer 0 (LPT1)

INT 17H

; AH contains return code

MOV AH, 0
; function 0, print character 

MOV AL, 0AH
; line feed

INT 17H




Interrupts 1Dh-1Fh 

These interrupt vectors point to data instead of instructions. The interrupt 1Dh, 1Eh, and 1Fh point to video initialization parameters, diskette parameters, and video graphics characters, respectively. 
DOS Interrupts[image: image15.png]



The interrupt types 20h-3Fh are serviced by DOS routines that provide high-level service to hardware as well as system resources such as files and directories. The most useful is INT 21H, which provides many functions for doing keyboard, video, and file operations. 
	20h
	Program Terminate
	Interrupt 20h can be used to return control to DOS. Because CS must be set to the program segment before using INT 20h, it is more convenient to use INT 21H, function 4Ch to exit.

	21h
	Function Request
	This routine provides over 80 functions that maybe classified as character I/O, file access, memory management, disk access, networking, and miscellaneous.

	22h 26h
	
	Ctrl -Break, critical errors, and direct disk access.

	27h
	Terminate but Stay Resident
	Allows programs to stay in memory after termination.


Interrupt 21h -- Function Request 

	0h
	Program Terminate
	Terminates the execution of a program

	1h
	Keyboard Input with Echo
	Read a character from the keyboard into AL with echo

	2h
	Display Output
	Display the character in DL to the screen

	5h
	Printer Output
	Outputs the character in DL to the printer

	8h
	Keyboard Input without Echo
	Read a character from the keyboard into AL without echo

	9h
	Print String
	Display the string characters addressed by DX to the screen

	0Ah
	Read String
	Read a string from the keyboard into buffer addressed by DX

	25h
	Set Vector
	Sets the address of an interrupt number in the IVT

	2Ah
	Get Date
	Returns the day of the week, year, month and date

	2Bh
	Set Date
	Sets the date

	2Ch
	Get Time
	Returns the time: hours, minutes, seconds, and hundredths

	2Dh
	Set Time
	Sets the time

	35h
	Get Vector
	Obtains the address of an interrupt number from the IVT

	39h
	Create Subdirectory (MKDIR)
	Creates the specified directory.

	3Ah
	Remove Subdirectory (RMDIR)
	Removes the specified directory

	3Bh
	Change Current Directory (CHDIR)
	Changes the current directory to the specified directory.

	3Ch
	Create a File
	Creates a new file

	3Dh
	Open a File
	Opens a file

	3Eh
	Close a File Handle
	Closes the specified file handle

	3Fh
	Read from a File
	Transfers the specified number of bytes from a file to a buffer

	40h
	Write to a File
	Transfers the specified number of bytes from a buffer into a file

	41h
	Delete a File from a Directory
	Removes a directory entry associated with a file name

	47h
	Get Current Directory
	Places the full path name of the current directory in the area pointed by DS:SI.

	48h
	Allocate Memory
	Allocates the requested number of paragraphs of memory

	49h
	Free Allocated Memory
	Frees the specified allocated memory

	4Ch
	Terminate a Process (EXIT)
	Terminates the current process and transfers control to the invoking process.


Next, we briefly describe some of the DOS INT 21h functions. 
	
	Function
	Input
	Output

	0BH
	Check keyboard buffer
	AH = 0BH
	AL = 00H - keyboard buffer is empty
AL = FFH - keyboard buffer is not empty

	0CH
	Clear keyboard buffer
	AH = 0BH .

AL = 01H, 06H, 07H, 08H, or 0AH
	Clears keyboard buffer and performs appropriate function depending on AL content

	05H
	Print a character
	AH = 05H 
DL = ASCII code of the character 
	None

	2AH

	Get Date
	AH = 2AH
	AL = Day of the week (0=SUN, 6=SAT)
CX = Year (1980-2099)

	2BH
	Set Date
	AH = 2BH
CX = Year (1980-2099)

DH = Month 

DL = Day (1-31)


	AL = 00h, if the date is valid
     FFh, if the date is not valid

	2CH
	Get Time
	AH = 2CH
	CH = Hours (0-23)
CL = Minutes 
DH = Seconds 

DL = Hundredths 



	2DH
	Set Time
	AH = 2DH
CH = Hours (0-23)
CL = Minutes
DH = Seconds
DL = Hundredths

	AL = 00h, if the time is valid
AL = FFh, if the time is not valid


Extended Keyboard Keys 

The IBM PC keyboard has several keys that are not the ASCII characters. These keys include the function keys, cursor arrows, Home, End, etc. These keys are called extended keys. When an extended key is pressed, the first byte placed in the keyboard buffer is 00H and the second byte is the keyboard scan code for the key. To read a character from the keyboard using DOS functions, extended keys require two function calls, as shown in the following procedure: 

	Read the next character code into AL using  INT 21h, function 08h
if (AL<>0) then

AL = ASCII code (ASCII character)
else

read the scan code of the extended key into AL using INT 21h, function 07h

AL = scan code (extended key character)
end if


A Time Display Program  [image: image16.png]



As an example of using interrupt routines, we write a program that displays the current time. When the computer is powered up, the current time can be entered by the user or supplied by a real-time clock circuit that is battery powered. This time value is kept in memory and updated by a timer circuit using interrupt 8. A program can call the DOS INT 21h, function 2Ch, to access the time. Our time display program has the following steps: 

· Obtain the current time 

· Convert the hours, minutes, and seconds into ASCII digits 

· Display the ASCII digits 

The procedure GET_TIME is used to get the current time and the procedure CONVERT is used to convert a number into ASCII. It first divides the number in AL by 10. This puts the ten's digit value in AL and the unit's digit value in AH (note that the input value is less than 60). Then, it converts both digits to ASCII. These two procedures are shown below: 

	Example: Procedures GET_TIME and CONVERT.

	PUBLIC GET_TIME
.MODEL SMALL
.CODE
GET_TIME PROC
        MOV AH, 2CH
; CH=hour, CL=min, DH=sec
        INT 21H
; convert hours into ASCII and store
        MOV AL, CH
        CALL CONVERT
        MOV [BX], AX
; convert minutes into ASCII and store
        MOV AL, CL
        CALL CONVERT
        MOV [BX+3], AX
; convert seconds into ASCII and store
        MOV AL, DH
        CALL CONVERT
        MOV [BX+6], AX
        RET
GET_TIME ENDP
CONVERT PROC
        XOR AH, AH
        MOV DL, 10
        DIV DL

; AH has remainder, AL has quotient
        OR AX, 3030H
; convert to ASCII
        RET
CONVERT ENDP
END


To display the current time, a time buffer, TIME_BUF, is initialized with 00:00:00. The procedure GET_TIME is then called to store the current time in the time buffer. Then, the time_buffer is displayed using INT 21h, function 9. This program is shown below: 

	Example: Display current time.

	EXTRN GET_TIME: NEAR
.MODEL SMALL
.STACK 100H
.DATA
TIME_BUF DB '00:00:00$';
.CODE
MAIN PROC
        MOV AX, @DATA

; initialize DS
        MOV DS, AX
; get and display time
        LEA BX, TIME_BUF
; BX points to TIME_BUF
        CALL GET_TIME

; put current time in TIME_BUF 
        LEA DX, TIME_BUF
; display time
        MOV AH, 9
        INT 21H
        MOV AH, 4CH

; return to dos
        INT 21H
MAIN ENDP
END MAIN


User Interrupt Procedures 

To make the time display program more interesting, let us write a second version that displays the time and updates it every second. One way to continuously update the time is to execute a loop that keeps obtaining the time via INT 21h, function 2Ch and displaying it. The problem here is to find a way to stop the program. 
Instead, this can be done by writing a routine for interrupt 1Ch. This interrupt is generated by the INT 8 routine which is activated by a timer circuit about 18.2 times a second. When our interrupt routine is called, it will get the time and display it. 

Set Interrupt Vector 

To set up an interrupt routine we need to do the following steps: 

· Save the current interrupt vector 

· Place the vector of the user procedure in the interrupt vector table 

· Restore the previous vector before termination the program 

We use the INT 21h, function 35h, to get the old vector and function 25h to set up the new interrupt vector. 

	Set interrupt Vector: INT 21h, function 25h
; store interrupt vector into vector table
Input:
AH = 25h

AL = interrupt  number

DS:DX = interrupt vector
Output: none 


	Get interrupt Vector: INT 21h, function 35h
; obtain interrupt vector from vector table
Input:
AH = 35h

AL = interrupt  number
Output: ES:BX = interrupt vector


The following procedure, SETUP_INT, saves an old interrupt vector and sets up a new vector. It gets the interrupt number in AL, a buffer to save the old vector at DS:DI, and a buffer containing the new interrupt vector at DS:SI. By reversing the two buffers, SETUP_INT can also be used to restore the old vector. 

	Example: Procedure SETUP_INT for setting interrupt vector.

	PUBLIC SETUP_INT
.MODEL SMALL
.CODE
SETUP_INT PROC
;saves old interrupt vector and sets a new one
;input: AL = interrupt number
;       DI = address of buffer for old vector
;       SI = address of buffer containing new vector 
; save old interrupt vector
        MOV AH, 35H     ;get vector
        INT 21H
        MOV [DI], BX    ;save offset
        MOV [DI+2], ES  ;save segment
;setup new vector
        MOV DX, [SI]
        PUSH DS
        MOV DS, [SI+2]
        MOV AH, 25H     ;set vector
        INT 21H
        POP DS          ;restore DS
        RET
SETUP_INT ENDP
END


Cursor Control 

Each display of the current time by INT 21h, function 9, will advance the cursor. If a new time is then displayed, it appears at a different screen position. So, to view the time updated at the same screen position we must restore the cursor to its original position before we display the time. This is achieved by determining the current cursor position; then, after each print string operation, we move the cursor back. We use INT 10h, functions 3 and 2, to save the original cursor position and to move the cursor to its original position after each print string operation. 

	Move Cursor: INT 10h, function 2
Input:
AH = 2

BH = page number

DH = row number

DL = column number
Output: none 


	Get Cursor Position: INT 10h, function 3
Input:
AH = 2

BH = page number
Output: DH = row number

DL = column number

CH = starting scan line for cursor

CL = ending scan line for cursor


Interrupt Procedure 

The interrupt procedure, TINE_INT, is written to perform the following steps: 

· Set DS 

· Get new time 

· Display time 

· Restore cursor position 

· Restore DS 

The main procedure is written to have the following steps: 

· Save the current cursor position 

· Set up the interrupt vector for the interrupt procedure, TIME_INT 

· Wait for a key input 

· Restore the old interrupt vector and terminate 

To do step 2, we use OFFSET and SEG to obtain the offset and segment of procedure SETUP_INT; the vector is then stored in the buffer NEW_VEC. The procedure SETUP_INT, is called to set up the vector for interrupt type 1Ch, timer tick. The interrupt 16h, function 0 is used for step 3, key input. Procedure SETUP_INT is again used in step 4; this time SI points to the old vector and DI points to the vector for TIME_INT. 
After setting up the cursor and interrupt vectors, the main procedure just waits for a keystroke. In the meantime, the interrupt procedure, TIME_INT, keeps updating the time whenever the timer circuit ticks. After a key is hit, the old interrupt vector is restored and the program terminates. The complete program is given below. 

	Example: Display updated time program.

	EXTRN GET_TIME:NEAR, SETUP_INT: NEAR
.MODEL SMALL
.STACK 100H
.DATA
TIME_BUF        DB '00:00:00$'
CURSOR_POS      DW ?
NEW_VEC         DW ?,?
OLD_VEC         DW ?,?
.CODE
MAIN PROC
        MOV AX, @DATA
        MOV DS, AX
;save cursor position
        MOV AH, 3
        MOV BH, 0
        INT 10H
        MOV CURSOR_POS, DX
; setup interrupt procedure
        MOV NEW_VEC, OFFSET TIME_INT
        MOV NEW_VEC+2, SEG TIME_INT
        LEA DI, OLD_VEC
        LEA SI, NEW_VEC
        MOV AL, 1CH
        CALL SETUP_INT
;read keyboard
        MOV AH, 0
        INT 16H
;restore old interrupt vector
        LEA DI, NEW_VEC
        LEA SI, OLD_VEC
        MOV AL, 1CH
        CALL SETUP_INT
;exit to DOS
        MOV AH, 4CH
        INT 21H
MAIN ENDP
;
TIME_INT PROC
        PUSH DS
        MOV AX, @DATA
        MOV DS, AX
; get new time
        LEA BX, TIME_BUF
        CALL GET_TIME
; display time
        LEA DX, TIME_BUF
        MOV AH, 9
        INT 21H
;restore cursor position
        MOV AH, 2
        MOV BH, 0
        MOV DX, CURSOR_POS
        INT 10H
        POP DS
        IRET
TIME_INT ENDP
END MAIN
END

In this section, we will discuss a group of INT 21h functions called file handle functions. We will describe what is meant by a file handle, file errors, opening and closing a file, reading from and writing to a file. 

File Handle 

When a file is opened or created in a program, DOS assigns it a unique number called the file handle. This number is used to identify the file, so the program must save it. There are five predefined file handles as shown in the following table: 

	File Handle 
	Device 

	0 
	keyboard 

	1 
	screen 

	2 
	error output -- screen 

	3 
	auxiliary device 

	4 
	printer 


File Errors 

There are many opportunities for errors in INT 21h file handling. DOS identifies each error by a code number. In the functions we describe here, if an error occurs then the CF is set and the code number appears in AX. The following table contains the most common file-handling errors: 

	Hex Error Code 
	Meaning 

	1 
	invalid function number 

	2 
	file not found 

	3 
	path not found 

	4 
	all available handles in use 

	5 
	access denied 

	6 
	invalid file handle 

	C 
	invalid access code 

	F 
	invalid drive specified 

	10 
	attempt to remove current directory 

	11 
	not the same device 

	12 
	no more files to be found 


Opening a New File

Before a file can be used, it must be opened. To create a new file or rewrite an existing file, the user provides a filename and an attribute and DOS returns a file handle. 

	Open a New File/Rewrite a File: INT 21h, function 3Ch
Input:
AH = 3Ch

DS:DX = address of file name, which is an ASCII string


 (a string ending with a 0 byte)

CL = attribute
Output:
if successful, AX = file handle

Error if CF = 1, error code in AX (3, 4, or 5)


The file name may include a path; for example C:\COE205\EX.ASM. Possible errors for this function are 3 (path does not exist), 4 (all file handles in use), or 5 (access denied, which means either that the directory is full or the file is read-only file). A file attribute is a byte that specifies some attributes for files as shown below: 

	Bit 
	Meaning if Set 

	0 
	Read-only file 

	1 
	Hidden file 

	2 
	DOS system file 

	3 
	Volume label 

	4 
	Subdirectory 

	5 
	Archive bit 

	6 
	Not used 

	7 
	Not used 


	Example: Write a program to open a new read-only file called FILE1.

	.model small
.stack 100h
.data
FNAME

DB
'FILE1', 0
HANDLE
DB
?
.code
.startup

MOV AH, 3Ch
; open file function

LEA DX, FNAME
; DX has filename address

MOV CL, 1
; read-only attribute

INT 21H

; open file

MOV HANDLE, AX
; save handle or error code

JC OPEN_ERROR
; jump if error

...
.exit
END


Opening an Existing File

To open an existing file, there is another function: 

	Open an Existing File: INT 21h, function 3Dh
Input:
AH = 3Dh

DS:DX = address of file name, which is an ASCII string


 (a string ending with a 0 byte)

AL = access code:
0 means open for reading




1 means open for writing




2 means open for both
Output:
if successful, AX = file handle

Error if CF = 1, error code in AX (2, 4, 5 or 12)


Closing a File

After a file is processed, it should be closed. This frees the file handle for use with another file. 

	Close a File: INT 21h, function 3Eh
Input:
BX = file handle
Output:
if CF = 1, error code in AX (6)


	Example: Write some code to close a file whose handle is stored in variable HANDLE.

	
MOV AX, 3Eh
; close file function

MOV BX, HANDLE
; get handle

INT 21H

; close file

JC CLOSE_ERROR
; jump if error

...


Reading from a File 

The following function reads a specified number of bytes from a file and stores them in memory. 

Read from a File: INT 21h, function 3Fh
Input:
AH = 3Fh

BX = file handle

CX = number of bytes to read

DS:DX = memory buffer address


Output:
AX = number of bytes actually read

if AX=0 or AX < CX, end of file encountered

Error if CF = 1, error code in AX (5, 6)
Example: Write some code to read a 512-byte from a file. Assume file handle is stored in variable HANDLE, and BUFFER is a 512 byte buffer.

.data


HANDLE DW
?


BUFFER DB 512 DUP(0)


...

.code


...


MOV AX, 3Fh
; read file function


MOV BX, HANDLE
; get handle


MOV CX, 512
; read 512 bytes


INT 21H

; read file, AX = bytes read


JC READ_ERROR
; jump if error


...

Writing to a File 

Function 40h writes a specified number of bytes to a file. 

	Write to a File: INT 21h, function 40h
Input:
AH = 40h

BX = file handle

CX = number of bytes to write

DS:DX = data address


Output:
AX = number of bytes written

if AX=0 or AX < CX, error (full disk)

Error if CF = 1, error code in AX (5, 6)


Function 40h writes data to a file, but it can also be used to send data to the screen or printer (handles 1 and 4 respectively). 

	Example: Use function 40h to display a message on the screen.

	.data

MSG DB 'Display This Message'

...
.code

...

MOV AX, 40h
; write file function

MOV BX, 1
; screen file handle

MOV CX, 20
; length of message

LEA DX, MSG
; get address of MSG

INT 21H

; display MSG

...

